



# MATEMÁTICA NA CULTURA

## RELIN/BAS: UM PROGRAMA GERAL DE REGRESSÃO LINEAR PARA MICROCOMPUTADORES DA LINHA APPLE

Ronaldo Sergio de Biasi

No presente trabalho é apresentado um programa em BASIC, desenvolvido para microcomputadores da linha Apple, que calcula e representa graficamente a reta que melhor se ajusta a um conjunto de até 100 pontos. Embora tenha sido desenvolvido com o objetivo final de preparar em uma impressora gráfica figuras para publicação em revistas científicas, o programa também pode ser usado para mostrar na tela do microcomputador os pontos experimentais e a reta de regressão, não sendo portanto indispensável para sua utilização que o usuário disponha de uma impressora.

### MÉTODO DE SOLUÇÃO

Dado um conjunto de pontos  $x_i, y_i$  ( $i = 1, 2, \dots, n$ ), os parâmetros da reta  $y = a + bx$  que melhor se ajustam aos pontos são dados por:<sup>1</sup>

$$b = \frac{\sum (x_i - \bar{x})(y_i - \bar{y})}{\sum (x_i - \bar{x})^2} = \frac{n \sum x_i y_i - \sum x_i \sum y_i}{n \sum x_i^2 - (\sum x_i)^2} \quad (1)$$

$$a = \bar{y} - b\bar{x} = (\sum y_i - b \sum x_i) / n \quad (2)$$

onde  $\bar{x} = (\sum x_i) / n$ ,  $\bar{y} = (\sum y_i) / n$

A qualidade do ajuste pode ser avaliada através de três parâmetros estatísticos: o coeficiente de determinação,  $r^2$ , o coeficiente de correlação,  $r$ , e o erro padrão da estimativa,  $s$ , definidos através das seguintes equações:<sup>1</sup>

$$r^2 = \frac{\sum (x_i - \bar{x})(y_i - \bar{y})}{\sum (x_i - \bar{x})^2 \sum (y_i - \bar{y})^2} = \frac{[\sum x_i y_i - (\sum x_i \sum y_i) / n]^2}{[\sum x_i^2 - (\sum x_i)^2 / n] [\sum y_i^2 - (\sum y_i)^2 / n]} \quad (3)$$

$$r = \sqrt{r^2} \quad (4)$$

$$s^2 = \frac{1}{n-2} \left\{ \sum (y_i - \bar{y})^2 - \frac{[\sum (x_i - \bar{x})(y_i - \bar{y})]^2}{\sum (x_i - \bar{x})^2} \right\} =$$

$$= \frac{1}{n-2} \left\{ \sum y_i^2 - \frac{(\sum y_i)^2}{n} - \frac{[\sum x_i y_i - (\sum x_i \sum y_i)/n]^2}{[\sum x_i^2 - (\sum x_i)^2/n][\sum y_i^2 - (\sum y_i)^2/n]} \right\} \quad (5)$$

$$s = \sqrt{s^2} \quad (6)$$

### UTILIZAÇÃO DO PROGRAMA

A listagem do programa aparece na Fig. 1. No início da execução, o programa mostra na tela um *menu* (Fig. 2) através do qual podem ser selecionadas as diferentes opções de entrada e saída. O *menu* é auto-explicativo. A seqüência normal de operação é a seguinte: a) entrar com os dados; b) calcular a regressão; c) entrar com os parâmetros do gráfico; d) ver o gráfico; e) imprimir o gráfico; f) imprimir os dados; g) imprimir o resultado da regressão. As opções auxiliares "ver os dados", "mudar os dados" e "mudar os parâmetros do gráfico" são úteis durante a fase de preparação do gráfico e podem ser usadas sempre que for conveniente. Se o usuário não dispuser de uma impressora, não deve usar as últimas três opções do *menu*. Caso, porém, selecione inadvertidamente uma delas, ainda terá a oportunidade de cancelar a escolha e voltar ao *menu*, bastando para isso apertar a tecla ESC.

### ESPECIFICAÇÕES DO PROGRAMA

- Linguagem: Applesoft BASIC.
- Memória utilizada: 13K bytes (aproximadamente).

- Entrada: pelo teclado.
- Saída: pelo vídeo e/ou pela impressora.
- Número máximo de pontos: 100.

```

10 GOSUB 280
20 X$ = "RELIN/BAS"
30 GOSUB 440
40 GOSUB 4050
50 X$ = "VOCE QUER:"
60 LM = 10
70 GOSUB 440
80 FOR I = 1 TO 10
90 PRINT TAB( 3);M$(I)
100 NEXT I
110 PRINT
120 VTAB 16: PRINT L$: PRINT
130 PRINT "USE A TECLA "; CHR$( 34); CHR$( PB); CHR$( 34);"
    PARA O CURSOR DESCER": PRINT
140 PRINT "USE A TECLA "; CHR$( 34); CHR$( PC); CHR$( 34);"
    PARA O CURSOR SUBIR": PRINT
150 PRINT "APERTE ";: INVERSE : PRINT
    "RETURN";: NORMAL : PRINT "
    PARA ESCOLHER"
160 GOSUB 550: IF TA ( ) 13 THEN
    GOTO 160
170 OP = L
180 X$ = M$(OP)
190 Z$ = "APERTE RETURN PARA VER
    MENU"
200 ON OP GOSUB 690,880,1220,141
    0,1790,2550,2700,2990,3250,3
    350
210 GOTO 50
220 PRINT "LIGUE A IMPRESSORA,":
    PRINT : PRINT "POSICIONE O
    PAPEL": PRINT
230 PRINT "NA PRIMEIRA LINHA DA
    PAGINA": PRINT
240 PRINT "E APERTE QUALQUER TEC
    LA": PRINT
250 PRINT "PARA COMECAR A IMPRIM
    IR.": PRINT
260 PRINT "PARA DESISTIR E VOLTA
    R AO MENU,": PRINT : PRINT "
    APERTE A TECLA ";: INVERSE :
    PRINT "ESC": NORMAL
270 PRINT : PRINT L$: RETURN
280 DIM C(100,2),M$(10),N$(3),P(
    100)
290 F = 0
300 F1 = 0:F2 = 0:F3 = 0:F4 = 0:F
    5 = 0:F6 = 0:F7 = 0

```

Figura 1 — Listagem do programa RELIN/BAS.

## MATEMÁTICA NA CULTURA

```

310 FOR I = 1 TO 100:P(I) = 0: NEXT
    I
320 L = 1
330 LI = 159:C0 = 279
340 DATA "ENTRAR COM OS DADOS",
    "VER OS DADOS","MUDAR OS DAD
    OS","CALCULAR A REGRESSAO","
    ENTRAR COM OS PARAMETROS DO
    GRAFICO"
350 DATA "VER O GRAFICO","MUDAR
    OS PARAMETROS DO GRAFICO","
    IMPRIMIR O GRAFICO","IMPRIMI
    R OS DADOS","IMPRIMIR O RESU
    LTADO DA REGRESSAO"
360 FOR I = 1 TO 10: READ M$(I):
    NEXT I
370 DATA "MODIFICAR PONTOS","AC
    RESCENTAR PONTOS","SUPRIMIR
    PONTOS"
380 FOR I = 1 TO 3: READ N$(I): NEXT
    I
390 PC = 83:PB = 68
400 OP = 1
410 L$ = "": FOR I = 1 TO 40:L$ =
    L$ + "*" : NEXT I
420 E$ = "": FOR I = 1 TO 40:E$ =
    E$ + " " : NEXT I
430 RETURN
440 HOME
450 XC = 21 - ( LEN (X$) / 2)
460 PRINT TAB(XC);X$
470 PRINT : PRINT L$
480 RETURN
490 VTAB 18: PRINT L$
500 VTAB 20: PRINT E$
510 VTAB 20: PRINT Y$
520 VTAB 22: PRINT E$
530 VTAB 22: PRINT Z$
540 RETURN
550 VTAB 4 + L: HTAB 1: PRINT " "
    ";
560 L1 = L
570 GOSUB 640
580 IF TA = PC AND L = 1 THEN L =
    LM + 1
590 IF TA = PC AND L > 1 THEN L =
    L - 1
600 IF TA = PB AND L = LM THEN L
    = 0
610 IF TA = PB AND L < LM THEN L
    = L + 1
620 IF L < > L1 THEN VTAB 4 +
    L1: HTAB 1: PRINT " " ;
630 RETURN
640 A$ = "": IF PEEK ( - 16384 ) >
    127 THEN GET A$
650 IF A$ = "" THEN GOTO 640
660 TA = ASC (A$)
670 RETURN
680 HTAB 16: PRINT "IMPRIMINDO"
690 X$ = "ENTRADA DE DADOS"
700 L = 2:F6 = 0:F7 = 0
710 HOME : GOSUB 450
720 PRINT "AO ENTRAR COM OS DADO
    S, NAO USE MAIS QUE"
730 PRINT " " 3 ALGARISMOS SI
    GNIFICATIVOS"
740 PRINT
750 PRINT
760 PRINT
770 PRINT "NUMERO DE PONTOS"
780 INPUT N
790 IF N < = 0 OR N > 100 THEN
    HOME : PRINT L$: PRINT "O V
    ALOR DE N DEVE ESTAR ENTRE 1
    E 100": PRINT : PRINT L$: GOSUB
    4180:N = 0:L = 1: RETURN
800 FOR I = 1 TO N:C(I,1) = 0:C(
    I,2) = 0: NEXT I
810 FOR I = 1 TO N
820 PRINT : PRINT "COORDENADAS (
    X,Y) DO PONTO ";I
830 PRINT "X=": INPUT X
840 PRINT "Y=": INPUT Y
850 C(I,1) = X:C(I,2) = Y
860 NEXT I
870 PRINT : PRINT "APERTE QUALQU
    ER TECLA PARA VER MENU": GET
    T$: RETURN
880 HOME : IF N = 0 THEN PRINT
    L$: HTAB 4: PRINT "VOCE NAO
    ENTROU COM NENHUM PONTO": PRINT
    : PRINT L$: GOSUB 4180:L = 1
    : RETURN
890 X$ = "DADOS": GOSUB 450
900 L = 2:A1 = 4:A2 = 18:A3 = 32
910 HTAB A1: PRINT "N": HTAB A2
    : PRINT "X": HTAB A3: PRINT
    "Y": PRINT
920 J = 0:Q = 0:Z1 = 19:Z2 = 33
930 FOR I = 1 TO N: HTAB A1: PRINT
    I: IF ABS (C(I,1)) > = 1E
    - 2 THEN HTAB Z1 - LEN ( STR$
    ( INT (C(I,1))))
940 IF ABS (C(I,1)) < 1E - 2 THEN
    HTAB Z1 + 4 - LEN ( STR$ (
    C(I,1)))
950 IF C(I,1) = 0 THEN HTAB Z1 -
    LEN ( STR$ (C(I,1)))
960 IF ABS (C(I,1)) < 1 AND ABS
    (C(I,1)) > = 1E - 2 AND C(I

```

Figura 1 – Listagem do programa RELIN/BAS (cont.).

```

,1) > 0 THEN PRINT "0";: PRINT
C(I,1);
970 IF ABS (C(I,1)) < 1E - 2 THEN
PRINT C(I,1);
980 IF ABS (C(I,1)) < 1 AND ABS
(C(I,1)) > = 1E - 2 AND C(I
,1) < 0 THEN PRINT "-0";:Q1
= - C(I,1): PRINT Q1;
990 IF ABS (C(I,1)) > = 1E9 THEN
HTAB Z1 + 4 - LEN ( STR$ (
C(I,1)))
1000 IF ABS (C(I,1)) > = 1 THEN
PRINT C(I,1);
1010 IF ABS (C(I,2)) < 1E - 2 THEN
HTAB Z2 + 4 - LEN ( STR$ (
C(I,2)))
1020 IF ABS (C(I,2)) > = 1E -
2 THEN HTAB Z2 - LEN ( STR$
( INT (C(I,2))))
1030 IF ABS (C(I,2)) < 1 AND ABS
(C(I,2)) > = 1E - 2 AND C(I
,2) > 0 THEN PRINT "0";: PRINT
C(I,2)
1040 IF C(I,2) = 0 THEN HTAB Z2
- LEN ( STR$ (C(I,2)))
1050 IF ABS (C(I,2)) < 1 AND ABS
(C(I,2)) > = 1E - 2 AND C(I
,2) < 0 THEN PRINT "-0";:Q2
= - C(I,2): PRINT Q2
1060 IF ABS (C(I,2)) > = 1E9 THEN
HTAB Z2 + 4 - LEN ( STR$ (
C(I,2)))
1070 IF ABS (C(I,2)) > = 1 THEN
PRINT C(I,2)
1080 IF ABS (C(I,2)) < 1E - 2 THEN
PRINT C(I,2)
1090 IF F3 = 1 THEN GOTO 1110
1100 J = J + 1: IF J = 10 THEN GOSUB
1160
1110 NEXT I
1120 IF F3 = 1 THEN GOTO 3330
1130 IF Q = 1 AND F3 = 0 THEN RETURN

1140 IF J = 10 THEN GOTO 1150
1150 IF I = N + 1 THEN Y$ = "APE
RTE QUALQUER TECLA PARA VER
O MENU":Z$ = "": GOSUB 490: GET
T$: RETURN
1160 Y$ = "APERTE QUALQUER TECLA
PARA CONTINUAR"
1170 Z$ = "APERTE RETURN PARA VOL
TAR AO MENU
1180 J = 0
1190 GOSUB 490
1200 GET T$:T = ASC (T$): IF T <
> 13 THEN FOR K = 1 TO 10:
UTAB K + 6: PRINT E$: NEXT
K: UTAB 7: RETURN
1210 I = N:Q = 1: RETURN
1220 X$ = "VOCE QUER:"
1230 F6 = 0:F7 = 0
1240 HOME
1250 GOSUB 450
1260 FOR I = 1 TO 3:
1270 PRINT TAB ( 3);N$(I)
1280 NEXT I
1290 PRINT
1300 Y$ = "USE " + CHR$ (34) + CHR$
(PC) + CHR$ (34) + " E " +
CHR$ (34) + CHR$ (PB) + CHR$
(34) + " PARA MOVIMENTAR O C
URSOR"
1310 Z$ = "APERTE RETURN PARA ESC
OLHER"
1320 GOSUB 490
1330 L = 1
1340 LM = 3
1350 GOSUB 550: IF TA ( ) 13 THEN
GOTO 1350
1360 OP = L
1370 X$ = N$(OP)
1380 Z$ = "APERTE RETURN PARA VER
MENU"
1390 ON OP GOSUB 3530,3690,3800
1400 RETURN

1410 HOME : IF N < 3 THEN PRINT
L$: HTAB 5: PRINT "NUMERO IN
SUFICIENTE DE PONTOS": PRINT
: PRINT L$:L = 2: GOSUB 4180
: RETURN
1420 X$ = "RESULTADO DA REGRESSAO
^": HOME : GOSUB 450: VTAB 5:
PRINT "EQUACAO DA RETA": PRINT

1430 G1 = 0:G2 = 0:G3 = 0:G4 = 0:
R = 0:F6 = 1
1440 FOR I = 1 TO N
1450 G1 = G1 + C(I,1)
1460 G2 = G2 + C(I,2)
1470 G3 = G3 + C(I,1) ^ 2
1480 G4 = G4 + C(I,2) ^ 2
1490 R = R + C(I,1) * C(I,2): NEXT
I
1500 B = (N * R - G2 * G1) / (N *
G3 - G1 ^ 2)
1510 A = (G2 - B * G1) / N
1520 L = 2
1530 IF ABS (A) < 1E - 8 THEN A
= 0
1540 IF ABS (B - 1) < 1E - 8 THEN
B = 1

```

Figura 1 – Listagem do programa RELIN/BAS (cont.).

## MATEMÁTICA NA CULTURA

```

1550 IF ABS ( B + 1 ) < 1E - 8 THEN
    B = - 1
1560 PRINT "F(X) =";
1570 IF A < > 0 THEN PRINT " "
    ;A;
1580 IF B = 1 THEN PRINT " + X"
    : GOTO 1620
1590 IF B = - 1 THEN PRINT " -
X": GOTO 1620
1600 IF B < 0 THEN PRINT " - ("
    ; ABS ( B ); " * X": GOTO 1620

1610 PRINT " + (<;B;" * X)"
1620 G1 = B * ( R - G1 * G2 / N )
1630 G4 = G4 - G2 ^ 2 / N
1640 G2 = G4 - G1
1650 IF G2 < 1E - 8 THEN G2 = 0
1660 PRINT : PRINT : PRINT
1670 IF G4 < 1E - 6 THEN R = 1: GOTO
    1690
1680 R = ABS ( G1 / G4 )
1690 PRINT "COEFICIENTE DE DETER
MINACAO: ";
1700 PRINT R
1710 PRINT
1720 PRINT "COEFICIENTE DE CORRE
LACAO: ";
1730 PRINT SQR ( R )
1740 PRINT
1750 PRINT "ERRO PADRAO DA ESTIM
ATIVA: ";
1760 PRINT SQR ( G2 / ( N - 2 ) )
1770 IF F4 = 1 THEN GOTO 3420
1780 VTAB 20: PRINT L$: GOSUB 41
80: RETURN
1790 X$ = "ENTRADA DOS PARAMETROS
DO GRAFICO": HOME : GOSUB 4
50
1800 L = 6:F7 = 0
1810 INVERSE : VTAB 12: PRINT E$
: VTAB 13: PRINT " PARA D
ESISTIR E VOLTAR AO MENU,
"
1820 VTAB 14: PRINT E$
1830 VTAB 15: PRINT " ENTR
E COM A PALAVRA ";: NORMAL :
PRINT "VOLTE";: INVERSE
1840 PRINT " "
1850 VTAB 16: PRINT E$: NORMAL
1860 VTAB 5
1870 PRINT "TITULO DO EIXO DOS X
";: INPUT W$
1880 IF W$ = "VOLTE" THEN RETURN
1890 FOR I = 1 TO 6: VTAB 11 + I
: PRINT E$: NEXT I

1900 VTAB 7
1910 PRINT "TITULO DO EIXO DOS Y
";: INPUT V$
1920 PRINT
1930 PRINT "VALOR MINIMO DE X, X
MIN";: INPUT X1: PRINT
1940 PRINT "VALOR MAXIMO DE X, X
MAX";: INPUT X2: PRINT
1950 IF X2 < = X1 THEN INVERSE
: PRINT "XMAX DEVE SER MAIOR
DO QUE XMIN": NORMAL : PRINT
: GOTO 1940
1960 PRINT "MARCAS NO EIXO DOS X
:"
1970 PRINT " VALOR INICIAL,XI
N";: INPUT X3: IF X3 < X1 OR
X3 > X2 THEN PRINT : INVERSE
: PRINT "ESCOLHA XIN TAL QUE
XMIN<=XIN<=XMAX": NORMAL : PRINT
: GOTO 1970
1980 PRINT " INCREMENTO, DX";
: INPUT DX: PRINT
1990 DX = ABS (DX): IF DX < (X2 -
X1) / 10 THEN INVERSE : PRINT
"INCREMENTO MUITO PEQUENO ":
NORMAL : PRINT : GOTO 1980
2000 PRINT "VALOR MINIMO DE Y, Y
MIN";: INPUT Y1: PRINT
2010 PRINT "VALOR MAXIMO DE Y, Y
MAX";: INPUT Y2: PRINT
2020 IF Y2 < = Y1 THEN INVERSE
: PRINT "YMAX DEVE SER MAIOR
QUE YMIN": NORMAL : PRINT :
GOTO 2010
2030 PRINT "MARCAS NO EIXO DOS Y
:"
2040 PRINT " VALOR INICIAL,YI
N";: INPUT Y3: IF Y3 < Y1 OR
Y3 > Y2 THEN PRINT : INVERSE
: PRINT "ESCOLHA YIN TAL QUE
YMIN<=YIN<=YMAX": NORMAL : PRINT
: GOTO 2040
2050 PRINT " INCREMENTO, DY";
: INPUT DY: PRINT
2060 DY = ABS (DY): IF DY < (Y2 -
Y1) / 10 THEN INVERSE : PRINT
"INCREMENTO MUITO PEQUENO ":
NORMAL : PRINT : GOTO 2050
2070 F2 = 1
2080 HOME : PRINT L$: HTAB 5: PRINT
"TERMINOU A ENTRADA DE PARAM
ETROS": PRINT : PRINT L$: GOSUB
4180: RETURN
2090 XESC = (X2 - X1) / C0:YESC =
(Y2 - Y1) / LI
2100 F = 0: FOR I = 1 TO N:P(I) =

```

Figura 1 – Listagem do programa RELIN/BAS (cont.).

```

0: NEXT I
2110 HOME = HGR2
2120 HPLOT 0,0 TO C0,0
2130 HPLOT C0,0 TO C0,LI
2140 HPLOT C0,LI TO 0,LI
2150 HPLOT 0,LI TO 0,0
2160 XX = X3
2170 R1 = INT ((XX - X1) / XESC)

2180 HPLOT R1,LI TO R1,LI - 4
2190 HPLOT R1,0 TO R1,4
2200 XX = XX + DX
2210 IF XX < X2 THEN GOTO 2170
2220 YY = Y3
2230 R2 = LI - INT ((YY - Y1) /
YESC)
2240 HPLOT 0,R2 TO 4,R2
2250 HPLOT C0,R2 TO C0 - 4,R2
2260 YY = YY + DY: IF YY < Y2 THEN
GOTO 2230
2270 IF F5 = 1 THEN GOTO 2350
2280 XI = 0:YI = LI - INT ((A +
X1 * B - Y1) / YESC)
2290 IF YI < 0 THEN GOTO 2450
2300 IF YI > LI THEN GOTO 2480
2310 XJ = C0:YJ = LI - INT ((A +
X2 * B - Y1) / YESC)
2320 IF YJ < 0 THEN GOTO 2510
2330 IF YJ > LI THEN GOTO 2540
2340 IF XI > = 0 AND XI < = C0
AND YI > = 0 AND YI < = L
I AND XJ > = 0 AND XJ < =
C0 AND YJ > = 0 AND YJ < =
LI THEN HPLLOT XI,YI TO XJ,Y
J
2350 IF N = 0 THEN GOTO 2430
2360 F5 = 0:I = 1
2370 Q1 = INT ((C(I,1) - X1) / X
ESC):Q2 = LI - INT ((C(I,2)
- Y1) / YESC)
2380 IF Q1 < 3 OR Q1 > C0 - 3 OR
Q2 < 3 OR Q2 > LI - 3 THEN GOTO
2410
2390 HPLLOT Q1,Q2 - 3 TO Q1,Q2 +
3: HPLLOT Q1 - 3,Q2 TO Q1 + 3
,Q2
2400 GOTO 2420
2410 P(I) = 1:F = 1
2420 I = I + 1: IF I < = N THEN
GOTO 2370
2430 GET T$: IF F = 1 THEN GOSUB
3960
2440 TEXT : RETURN
2450 XI = ((Y2 - A) / B - X1) / X
ESC
2460 YI = 0

2470 GOTO 2310
2480 XI = ((Y1 - A) / B - X1) / X
ESC
2490 YI = LI
2500 GOTO 2310
2510 XJ = ((Y2 - A) / B - X1) / X
ESC
2520 YJ = 0
2530 GOTO 2340
2540 XJ = ((Y1 - A) / B - X1) / X
ESC:YJ = LI: GOTO 2340
2550 IF N = 0 THEN GOSUB 880:L =
1: RETURN
2560 IF F2 = 0 THEN HOME : PRINT
L$: HTAB 4: PRINT "PRIMEIRO
ENTRE COM OS PARAMETROS": PRINT
: PRINT L$:L = 5: GOSUB 4180
: RETURN
2570 IF X2 < = X1 OR Y2 < = Y1
OR X3 < X1 OR X3 > X2 OR Y3
< Y1 OR Y3 > Y2 OR DX < (X2
- X1) / 10 OR DY < (Y2 - Y1
) / 10 THEN GOSUB 4250: RETURN
2580 L = 2: HOME : PRINT L$: VTAB
5: PRINT L$: VTAB 3: PRINT "
COM RETA DE REGRESSAO (S/N)"
: INPUT R$
2590 IF R$ = "N" THEN F5 = 1: GOTO
2610
2600 IF R$ < > "S" THEN GOTO 2
580

2610 IF F6 = 0 THEN HOME : PRINT
L$: HTAB 6: PRINT "PRIMEIRO
CALCULE REGRESSAO": PRINT : PRINT
L$: GOSUB 4180: RETURN
2620 HOME : VTAB 5: PRINT L$
2630 PRINT "APERTE QUALQUER TECL
A"
2640 PRINT : PRINT "PARA VER O G
RAFICO"
2650 PRINT : PRINT
2660 PRINT "DEPOIS DE VER O GRAF
ICO"
2670 PRINT : PRINT "APERTE QUALQ
UER TECLA"
2680 PRINT : PRINT "PARA VOLTAR
AO MENU": PRINT : PRINT L$
2690 F7 = 1: GET T$: GOTO 2090
2700 X$ = "MUDAR PARAMETROS"
2710 HOME : GOSUB 450
2720 L = 6:F7 = 0
2730 PRINT "NO MOMENTO, OS PARAM
ETROS SAO:"
2740 PRINT

```

Figura 1 – Listagem do programa RELIN/BAS (cont.).

## MATEMÁTICA NA CULTURA

```

2750 PRINT "TITULO DO EIXO DOS X
: ";: PRINT W$
2760 PRINT : PRINT "TITULO DO EI
XO DOS Y: ";: PRINT V$
2770 PRINT
2780 PRINT "EIXO DOS X
EIXO DOS Y"
2790 PRINT "-----
-----"
2800 PRINT "XMIN: ";X1;: HTAB 22
: PRINT "YMIN: ";Y1
2810 PRINT "XMAX: ";X2;: HTAB 22
: PRINT "YMAX: ";Y2
2820 PRINT "XIN: ";X3;: HTAB 22
: PRINT "YIN: ";Y3: PRINT "
DX: ";DX;: HTAB 22: PRINT
"DY: ";DY
2830 IF F1 = 1 THEN GOTO 3230
2840 PRINT : PRINT L$: PRINT "EN
TRE COM O NOME E O VALOR DO
PARAMETRO"
2850 PRINT " (POR EXEMPLO
: XMAX,20)": PRINT
2860 Z$ = "QUANDO TERMINAR, ENTRE
COM FIM,0"
2870 PRINT "QUANDO TERMINAR, ENT
RE COM FIM,0
2880 INPUT P$,P
2890 IF P$ = "XMAX" THEN X2 = P:
GOTO 2710
2900 IF P$ = "XMIN" THEN X1 = P:
GOTO 2710
2910 IF P$ = "YMAX" THEN Y2 = P:
GOTO 2710
2920 IF P$ = "YMIN" THEN Y1 = P:
GOTO 2710
2930 IF P$ = "XIN" THEN X3 = P: GOTO
2710
2940 IF P$ = "YIN" THEN Y3 = P: GOTO
2710
2950 IF P$ = "DX" THEN DX = P: GOTO
2710
2960 IF P$ = "DY" THEN DY = P: GOTO
2710
2970 IF P$ = "FIM" THEN HOME : RETURN
2980 VTAB 23: PRINT "PARAMETRO D
ESCONHECIDO": FOR W = 1 TO 7
00: NEXT W: GOTO 2710
2990 IF F7 = 0 THEN HOME : PRINT
L$: HTAB 9: PRINT "PRIMEIRO
VEJA O GRAFICO": PRINT : PRINT
L$: GOSUB 4180:L = 6: RETURN
3000 X$ = "IMPRESSAO DO GRAFICO":
HOME : GOSUB 450: GOSUB 301
0: GOTO 3080
3010 PRINT "LIGUE A IMPRESSORA E
": PRINT : PRINT "POSICIONE
O PAPEL": PRINT
3020 PRINT "NA PRIMEIRA LINHA DA
PAGINA."
3030 PRINT : PRINT : PRINT "PARA
IMPRIMIR:"
3040 PRINT : HTAB 5: PRINT "EM T
AMANHO NORMAL, ENTRE COM ";:
INVERSE : PRINT "N": NORMAL
3050 PRINT : HTAB 5: PRINT "EM T
AMANHO REDUZIDO,ENTRE COM ";
: INVERSE : PRINT "R": NORMAL
3060 PRINT : PRINT "PARA DESISTI
R E VOLTAR AO MENU,": PRINT
: HTAB 5: PRINT "APERTE A TE
CLA ";: INVERSE : PRINT "ESC
": NORMAL
3070 PRINT : PRINT L$: RETURN
3080 GET T$:AS = ASC (T$): IF A
S ( ) 27 AND AS ( ) 78 AND
AS ( ) 82 THEN GOTO 3080
3090 IF ASC (T$) = 27 THEN L =
2: RETURN
3100 GOSUB 3450: PR# 1
3110 IF ASC (T$) = 82 THEN PRINT
CHR$(9);"G2": GOTO 3140
3120 IF ASC (T$) = 78 THEN PRINT
CHR$(9);"GD2": GOTO 3140
3130 PR# 0: RETURN
3140 IF F = 1 THEN PRINT "NAO F
OI POSSIVEL MARCAR O(S) PONT
O(S)"
3150 FOR I = 1 TO N: IF P(I) = 1
THEN PRINT I;" ";
3160 NEXT I
3170 PRINT : PRINT
3180 PRINT CHR$(9);"40N"
3190 PRINT L$: PRINT
3200 F1 = 1
3210 L = 2
3220 GOTO 2750
3230 PRINT : PRINT : PRINT L$
3240 F1 = 0: PR# 0: RETURN
3250 IF N > 0 THEN GOTO 3270
3260 GOSUB 880:L = 1: RETURN
3270 HOME : VTAB 5: PRINT L$: GOSUB
220
3280 GET T$:AS = ASC (T$): IF A
S = 27 THEN L = 2: RETURN
3290 F3 = 1
3300 PR# 1
3310 GOTO 890
3320 GOSUB 220

```

Figura 1 – Listagem do programa RELIN/BAS (cont.).

```

3330 PRINT L$
3340 PR# 0:F3 = 0: RETURN
3350 HOME
3360 L = 2
3370 IF F6 = 0 THEN PRINT L$: HTAB
      B: PRINT "PRIMEIRO CALCULE R
      EGRESSAO": PRINT : PRINT L$:
      GOSUB 4180:L = 4: RETURN
3380 VTAB 5: PRINT L$: GOSUB 220

3390 F4 = 1
3400 GET T$:AS = ASC (T$): IF A
      S = 27 THEN L = 2: RETURN
3410 PR# 1: GOTO 1420
3420 PRINT L$
3430 F4 = 0
3440 PR# 0: RETURN
3450 HOME : VTAB 5: PRINT L$
3460 PRINT
3470 FLASH
3480 HTAB 16: PRINT "IMPRIMINDO"

3490 NORMAL
3500 PRINT : PRINT
3510 PRINT L$
3520 RETURN
3530 X$ = N$(1)
3540 L = 2
3550 HOME : GOSUB 450
3560 PRINT "ENTRE COM O NUMERO E
      AS COORDENADAS"
3570 PRINT
3580 PRINT "DO PONTO A SER MODIF
      ICADO": PRINT : PRINT
3590 PRINT "ENTRE COM I>100 PARA
      VOLTAR AO MENU"
3600 PRINT
3610 PRINT "I=";: INPUT I
3620 PRINT
3630 IF I > 100 THEN RETURN
3640 IF I > N THEN PRINT "ESTE
      PONTO NAO EXISTE!": FOR K =
      1 TO 1000: NEXT K: GOTO 3600

3650 PRINT "X=";: INPUT X
3660 PRINT "Y=";: INPUT Y
3670 C(I,1) = X:C(I,2) = Y
3680 PRINT : GOTO 3530
3690 X$ = N$(2)
3700 HOME : GOSUB 450
3710 PRINT "ENTRE COM AS COORDEN
      ADAS DO PONTO": PRINT
3720 PRINT "ENTRE COM X=999 PARA
      VOLTAR AO MENU"
3730 PRINT : PRINT : PRINT "COOR
      DENADAS DO PONTO ";N + 1

3740 PRINT : PRINT "X=";: INPUT
      X
3750 IF X = 999 THEN RETURN
3760 PRINT "Y=";: INPUT Y
3770 N = N + 1
3780 C(N,1) = X:C(N,2) = Y
3790 GOTO 3690
3800 X$ = N$(3)
3810 L = 2
3820 HOME : GOSUB 450
3830 IF N = 1 THEN INVERSE : VTAB
      5: PRINT E$: PRINT E$: PRINT
      E$: VTAB 6: PRINT E$: PRINT
      E$: VTAB 6: HTAB 8: PRINT "V
      OCE SO TEM UM PONTO...": VTAB
      8: HTAB 9: PRINT "NAO DEVE S
      UPRIMI-LO!": NORMAL : FOR K =
      1 TO 1000: NEXT K: RETURN
3840 IF N = 100 THEN PRINT : PRINT
      "VOCE JA TEM 100 PONTOS...":
      FOR K = 1 TO 1000: NEXT K: RETURN
3850 PRINT "ENTRE COM O NUMERO D
      O PONTO": PRINT : PRINT "A S
      ER SUPRIMIDO": PRINT : PRINT

3860 PRINT "I=";: INPUT I
3870 IF I > N THEN PRINT : PRINT
      "ESTE PONTO NAO EXISTE!": FOR
      K = 1 TO 1000: NEXT K: RETURN

3880 FOR J = I TO N
3890 C(J,1) = C(J + 1,1)
3900 C(J,2) = C(J + 1,2)
3910 NEXT J
3920 C(N,1) = 0:C(N,2) = 0
3930 N = N - 1
3940 X$ = "NOVOS DADOS": HOME : GOSUB
      450
3950 POP : GOTO 900
3960 HOME : TEXT : VTAB 5
3970 PRINT L$: PRINT "NAO FOI PO
      SSIVEL MARCAR 0(S) PONTO(S)"

3980 FOR I = 1 TO N: IF P(I) = 1
      THEN PRINT I;" ";
3990 NEXT I
4000 PRINT : PRINT
4010 PRINT "APERTE QUALQUER TECL
      A PARA VER MENU"
4020 PRINT : PRINT L$
4030 GET T$
4040 RETURN
4050 VTAB 5
4060 HTAB 5
4070 PRINT "UM PROGRAMA DE REGRE
      SSAO LINEAR"

```

Figura 1 – Listagem do programa RELIN/BAS (cont.).



```

4080 PRINT
4090 HTAB 13
4100 PRINT "DESENVOLVIDO POR"
4110 PRINT
4120 HTAB 9
4130 PRINT "RONALDO SERGIO DE BI
ASI"
4140 PRINT
4150 HTAB 16
4160 PRINT "IME/1986"
4170 PRINT : PRINT L$: GOSUB 418
Ø: RETURN
4180 VTAB 20
4190 INVERSE : PRINT E$: VTAB 21

4200 PRINT " APERTE QUALQUER TEC
LA PARA VER O MENU "
4210 VTAB 22: PRINT E$: NORMAL
4220 T$ = "": IF PEEK ( - 16384)
) 127 THEN GET T$
4230 IF T$ = "" THEN GOTO 4220
4240 RETURN
4250 HOME : PRINT L$: PRINT "UM
DOS PARAMETROS DO GRAFICO ES
TA ERRADO"
4260 HTAB 11: PRINT "SE NAO SOUB
ER QUAL," : PRINT
4270 PRINT " ENTRE DE NOVO COM T
ODOS OS PARAMETROS": PRINT :
PRINT L$: GOSUB 4180:L = 7:
RETURN
    
```

Figura 1 – Listagem do programa RELIN/BAS (cont.).

### EXEMPLO

Um exemplo típico, que serve para ilustrar o formato de saída, aparece nas Figs. 3 a 5. No exemplo, que foi tirado de um problema real,<sup>2</sup> é determinada a reta que melhor se ajusta a um conjunto de valores de largura de linha de ressonância paramagnética eletrônica em função da concentração de cromo em amostras de óxido de magnésio.

### PROGRAMA EM DISQUETE

Para obter uma cópia deste programa em disquete, envie um disquete virgem de 5 1/4" e um envelope selado e auto-endereçado para: Dr. Ronaldo Sergio de Biasi, Seção de Engenharia e Ciência dos Materiais, Instituto Militar de Engenharia, 22290 Rio de Janeiro, RJ. Em menos de um mês, receberá de volta o seu disquete com o programa gravado.

DADOS

```

*****
N          X          Y
*****
1          0.03        3.58
2          0.07        3.99
3          0.13        4.14
4          0.2         5.05
5          0.27        5.2
6          0.33        5.73
7          0.4         6.12
8          0.47        6.52
9          0.53        7.45
10         0.6         8.01
11         0.66        8.18
12         0.73        8.39
13         0.79        9.08
14         0.99        10.29
*****
    
```

Figura 2 – Menu principal do programa.

VOCE QUER:

```

*****
) ENTRAR COM OS DADOS
  VER OS DADOS
  MUDAR OS DADOS
  CALCULAR A REGRESSAO
  ENTRAR COM OS PARAMETROS DO GRAFICO
  VER O GRAFICO
  MUDAR OS PARAMETROS DO GRAFICO
  IMPRIMIR O GRAFICO
  IMPRIMIR OS DADOS
  IMPRIMIR O RESULTADO DA REGRESSAO
*****
USE A TECLA "D" PARA O CURSOR DESCER
USE A TECLA "S" PARA O CURSOR SUBIR
APERTE RETURN PARA ESCOLHER
    
```

Figura 3 – Exemplo de utilização do programa: dados de entrada (extraídos da referência 2).

RESULTADO DA REGRESSAO

\*\*\*\*\*  
EQUACAO DA RETA

$$F(X) = 3.4099288 + (7.09532207 * X)$$

COEFICIENTE DE DETERMINACAO: .991963699

COEFICIENTE DE CORRELACAO: .995973744

ERRO PADRAO DA ESTIMATIVA: .192765155

\*\*\*\*\*

Figura 4 – Exemplo de utilização do programa: resultado da regressão.


AGRADECIMENTO

O autor agradece ao Prof. Antonio Alberto Ribeiro Fernandes, que testou exaustivamente o programa, ajudando assim a localizar e corrigir muitos erros e omissões.

REFERÊNCIAS BIBLIOGRÁFICAS

1. *HP Model 10 Math Pac*, Hewlett-Packard, Loveland, 1971, p. 167.
2. BIASI, R.S. de, e RODRIGUES, D.C.S., *Influence of chromium concentration and particle size on the ESR line-width of Al<sub>2</sub>O<sub>3</sub>:Cr<sup>3+</sup> powders*, *J.Mater.Sci* 16, 968 (1981).

**Nota.** Breve biografia do Prof. Ronaldo Sérgio de Biasi encontra-se no artigo "O modelo de Kronig-Penney", neste mesmo número da RMCT.



\*\*\*\*\*  
TITULO DO EIXO DOS X: F(MOL%)  
TITULO DO EIXO DOS Y: DHPP(MILITESLA)  
EIXO DOS X                      EIXO DOS Y  
-----  
XMIN: 0                            YMIN: 0  
XMAX: 1.25                        YMAX: 15  
XIN: 0                              YIN: 0  
DX: .25                             DY: 2.5  
\*\*\*\*\*

Figura 5 – Exemplo de utilização do programa: gráfico mostrando os pontos experimentais e a reta de regressão.